# Maintainer: Vadim Yanitskiy

pkgname=gr-gsm
pkgver=v1165.13e42c4
pkgrel=1
pkgdesc="GNU Radio blocks and tools for receiving GSM transmissions"
arch=('any')
url="https://github.com/ptrkrysik/gr-gsm"
license=('GPL2')
# TODO: libosmocore?
depends=('gnuradio' 'gnuradio-osmosdr' 'boost' 'swig')
makedepends=('cmake')
provides=('gr-gsm')
source=(git://git.osmocom.org/${pkgname})
sha1sums=('SKIP')

pkgver() {
  cd "$srcdir/$pkgname"
  printf "v%s.%s" "$(git rev-list --count HEAD)" "$(git rev-parse --short HEAD)"
}

build() {
  cd "$srcdir/$pkgname"
  mkdir build && cd build
  cmake \
    -DCMAKE_INSTALL_PREFIX=/usr \
    -DENABLE_DOXYGEN=OFF \
    -Wno-dev \
    ..
  make
}

package() {
  cd "$srcdir/$pkgname/build"
  make DESTDIR=${pkgdir} install
}

# vim:set ts=2 sw=2 et:
